

9/13/99

A gloss on Gould's section 10.2.1

Here we address the question of edge moments from the point of view of the biharmonic equation, to obtain an analysis complementary to that of Gould.

Look at the **homogeneous solution to the biharmonic equation** in Cartesian coordinates, which comes out in polynomials. Put in all possible terms

```
STUDENT > a:=array(0..3,0..3):  
            w:=0:  
            for i from 0 to 3 do for j from 0 to 3 do  
            w:=w+a[i,j]*x^i*y^j od:od:
```

The Laplacian of w

```
STUDENT > Dw:=diff(diff(w,x),x)+diff(diff(w,y),y);  
Dw := 2 a2,0 + 2 a2,1 y + 2 a2,2 y2 + 2 a2,3 y3 + 6 a3,0 x + 6 a3,1 x y + 6 a3,2 x y2 + 6 a3,3 x y3  
      + 2 a0,2 + 6 a0,3 y + 2 a1,2 x + 6 a1,3 x y + 2 a2,2 x2 + 6 a2,3 x2 y + 2 a3,2 x3 + 6 a3,3 x3 y
```

Its Laplacian

```
STUDENT > DDw:=diff(diff(Dw,x),x)+diff(diff(Dw,y),y);  
DDw := 8 a2,2 + 24 a2,3 y + 24 a3,2 x + 72 a3,3 x y
```

Conditions for a solution. (If the right hand side is constant, then we need merely retain a[2,2].)

```
STUDENT > a[2,2]:=0:a[2,3]:=0:a[3,2]:=0:a[3,3]:=0:w;  
a0,0 + a0,1 y + a0,2 y2 + a0,3 y3 + a1,0 x + a1,1 x y + a1,2 x y2 + a1,3 x y3 + a2,0 x2 + a2,1 x2 y + a3,0 x3  
+ a3,1 x3 y
```

The general homogeneous deflection of a plate then is given by:

```
STUDENT > w;  
a0,0 + a0,1 y + a0,2 y2 + a0,3 y3 + a1,0 x + a1,1 x y + a1,2 x y2 + a1,3 x y3 + a2,0 x2 + a2,1 x2 y + a3,0 x3  
+ a3,1 x3 y
```

The book treats **moment loads**, and we can look at that here. Suppose the boundaries to be at $x = \pm Lx/2$ and $y = \pm Ly/2$, and let the moments at these locations be Mxm , Mxp , Mym , Myp , respectively. We ought to be able to find values of the coefficients that will match these imposed conditions. Note that we cannot impose zero deflection on the boundaries for this problem, so it is somewhat artificial

```
STUDENT > Mx:=-D*(diff(diff(w,x),x)+nu*diff(diff(w,y),y));  
My:=-D*(diff(diff(w,y),y)+nu*diff(diff(w,x),x));  
  
Mx := -D (2 a2,0 + 2 a2,1 y + 6 a3,0 x + 6 a3,1 x y + v (2 a0,2 + 6 a0,3 y + 2 a1,2 x + 6 a1,3 x y))  
My := -D (2 a0,2 + 6 a0,3 y + 2 a1,2 x + 6 a1,3 x y + v (2 a2,0 + 2 a2,1 y + 6 a3,0 x + 6 a3,1 x y))
```

We want to look at the response to various edge loadings. In all cases we can look at zero

displacement boundary conditions, and I will let the boundaries be at $\pm L1/2$ and $\pm L2/2$. so we have to being with

```
STUDENT > x:=-L1/2:e1:=Mx-Mxm;x:=L1/2:e2:=Mx-Mxp;
           x:='x':
           y:=-L2/2:e3:=My-Mym;y:=L2/2:e4:=My-Myp;
           y:='y':
```

```
STUDENT > e1:=collect(e1,y);e2:=collect(e2,y);
           e3:=collect(e3,x);e4:=collect(e4,x);
```

```
e1 := -D(2 a2,1 - 3 a3,1 L1 + v(6 a0,3 - 3 a1,3 L1)) y
      - D(2 a2,0 - 3 a3,0 L1 + v(2 a0,2 - a1,2 L1)) - Mxm
e2 := -D(2 a2,1 + 3 a3,1 L1 + v(6 a0,3 + 3 a1,3 L1)) y
      - D(2 a2,0 + 3 a3,0 L1 + v(2 a0,2 + a1,2 L1)) - Mxp
e3 := -D(2 a1,2 - 3 a1,3 L2 + v(6 a3,0 - 3 a3,1 L2)) x
      - D(2 a0,2 - 3 a0,3 L2 + v(2 a2,0 - a2,1 L2)) - Mym
e4 := -D(2 a1,2 + 3 a1,3 L2 + v(6 a3,0 + 3 a3,1 L2)) x
      - D(2 a0,2 + 3 a0,3 L2 + v(2 a2,0 + a2,1 L2)) - Myp
```

The applied moments are constant, so the nonconstant parts of these must vanish

```
STUDENT > e11:=coeff(e1,y);e21:=coeff(e2,y);
           e31:=coeff(e3,x);e41:=coeff(e4,x);
```

```
e11 := -D(2 a2,1 - 3 a3,1 L1 + v(6 a0,3 - 3 a1,3 L1))
e21 := -D(2 a2,1 + 3 a3,1 L1 + v(6 a0,3 + 3 a1,3 L1))
e31 := -D(2 a1,2 - 3 a1,3 L2 + v(6 a3,0 - 3 a3,1 L2))
e41 := -D(2 a1,2 + 3 a1,3 L2 + v(6 a3,0 + 3 a3,1 L2))
```

```
STUDENT > solve({e11,e21,e31,e41},{a[1,2],a[2,1],a[1,3],a[3,1]});
           {a1,2=-3 v a3,0, a3,1=0, a2,1=-3 v a0,3, a1,3=0}
```

```
STUDENT > assign("):e1;e2;e3;e4;
```

```
-D(2 a2,0 - 3 a3,0 L1 + v(2 a0,2 + 3 v a3,0 L1)) - Mxm
-D(2 a2,0 + 3 a3,0 L1 + v(2 a0,2 - 3 v a3,0 L1)) - Mxp
-D(2 a0,2 - 3 a0,3 L2 + v(2 a2,0 + 3 v a0,3 L2)) - Mym
-D(2 a0,2 + 3 a0,3 L2 + v(2 a2,0 - 3 v a0,3 L2)) - Myp
```

```
STUDENT > solve({e1,e2,e3,e4},{a[2,0],a[0,2],a[3,0],a[0,3]});
```

```
{a0,3 = -1/6 * (-Myp + Mym) / (D L2 (-1 + v^2)), a3,0 = 1/6 * (Mxp - Mxm) / (D L1 (-1 + v^2)), a2,0 = 1/4 * (-v Myp - v Mym + Mxm + Mxp) / (D (-1 + v^2)),
a0,2 = -1/4 * (-Myp - Mym + v Mxm + v Mxp) / (D (-1 + v^2)) }
```

```
STUDENT > assign("):
```

```
STUDENT > w;
```

```
a0,0 + a0,1 y - 1/4 * (-Myp - Mym + v Mxm + v Mxp) y^2 / (D (-1 + v^2)) - 1/6 * (-Myp + Mym) y^3 / (D L2 (-1 + v^2)) + a1,0 x + a1,1 x y
- 1/2 * v (Mxp - Mxm) x y^2 / (D L1 (-1 + v^2)) + 1/4 * (-v Myp - v Mym + Mxm + Mxp) x^2 / (D (-1 + v^2)) + 1/2 * v (-Myp + Mym) x^2 y / (D L2 (-1 + v^2))
```

$$+ \frac{1}{6} \frac{(M_{xp} - M_{xm}) x^3}{D L I (-1 + \nu^2)}$$

The text result has only quadratic terms. If $M_{xm} = M_{xp}$ and $M_{ym} = M_{yp}$, the cubic terms vanish. If the moments are not equal, then the system is not in equilibrium without the addition of forces, and these forces would determine $a[1,0]$, $a[0,1]$ and $a[1,1]$. We can take **moment equality and symmetry**, let the origin of the z coordinate be at $x = 0 = y$, set all the extraneous stuff = 0, and change notation a bit to make the result look exactly like Gould's.

```
STUDENT > a[0,0]:=0:a[1,0]:=0:a[0,1]:=0:a[1,1]:=0:
Mx:='Mx':My:='My':Mxp:=M1:Mxm:=M1:Mym:=M2:Myp:=M2:
nu:=mu:
factor(op(1,w))+factor(op(2,
w));
```

$$-\frac{1}{2} \frac{(-M2 + \mu MI) y^2}{D (\mu - 1) (\mu + 1)} + \frac{1}{2} \frac{(-\mu M2 + MI) x^2}{D (\mu - 1) (\mu + 1)}$$

which is exactly the same as Gould's equation 10.64. Note that $L1$ and $L2$ do not enter directly in the solution.

```
STUDENT > W:="";
```

$$W := -\frac{1}{2} \frac{(-M2 + \mu MI) y^2}{D (\mu - 1) (\mu + 1)} + \frac{1}{2} \frac{(-\mu M2 + MI) x^2}{D (\mu - 1) (\mu + 1)}$$

Now we can look at **special cases** after we've defined an amplitude function for the sake of plotting surfaces

```
STUDENT > plotw:=factor(W*D*(1-mu^2));
```

$$plotw := -\frac{1}{2} y^2 M2 + \frac{1}{2} y^2 \mu MI + \frac{1}{2} x^2 \mu M2 - \frac{1}{2} x^2 MI$$


1. Case 1: **parabolic bending** $M1 = M = M2$

```
STUDENT > M1:=M:M2:=M:factor(plotw);
```

$$\frac{1}{2} M (y^2 + x^2) (\mu - 1)$$

```
STUDENT > plot3d({0,(x^2+y^2)/10},x=-1..1,y=-2..2,axes=NORMAL,shading=Z,
title=`Parabolic surface with aspect ratio 2:1`);
```

Parabolic surface with aspect ratio 2:1




Reset everything for case 2, **cylindrical bending**

```
STUDENT > M1:='M1':M2:='M2':M2:=M:M1:=mu*M:factor(plotw);
```

$$\frac{1}{2}y^2 M (\mu - 1) (\mu + 1)$$

```
STUDENT > plot3d({0,(y^2)/10},x=-1..1,y=-2..2,axes=NORMAL,shading=Z,  
title=`Cylindrical surface with aspect ratio 2:1`);
```

Cylindrical surface with aspect ratio 2:1




And finally, we can look at case 3, **anticlastic bending**

```
STUDENT > M1:=M:M2:=-M:factor(plotw);
```

$$-\frac{1}{2}M(x-y)(x+y)(\mu+1)$$

```
STUDENT > plot3d({0,(-x^2+y^2)/10},x=-1..1,y=-2..2,axes=NORMAL,shading=Z,  
title=`Anticlastic surface with aspect ratio 2:1`);
```

Anticlastic surface with aspect ratio 2:1



I save a computation file in case I want to come back to this

```
STUDENT > save `homopolynom.m`:
```